

# MikroASSEMBLER

## Apple II

Ennen kuin syötät listauksen koneeseesi, tutustu ohjelmien syöttöohjeisiin kirjan liitteestä.

Applen MikroASSEMBLER toimii pääpiirteiltään kuten aiemmin on selostettu. Kuitenkin eräitä ominaisuuksia on muutettu. Ohjelma on suunniteltu tallentamaan ohjelmia pelkästään levyille, joten kasettioperaatioita ei voi käynnistää. Pistekoodit ovat samat, mutta disassemblerin sekä monitorin ohjausnäppäimiä on muutettu.

Ollessasi monitorissa, siirryt tavun eteenpäin painamalla RETURNia. Taaksepäin pääsee TABilla (Control-I, jollei TAB-näppäintä ole). Monitorista poistutaan painamalla ESC-näppäintä. Disassemblauksesta pääsee pois X-näppäimen painalluksella.

```

1  REM  **** MIKRO ASSEMBLER ****
2  REM
3  REM  converted to Apple II by
4  REM      Aki Korhonen
5  REM
15 H$ = CHR$ (8): REM  BACKSPACE
16 D$ = CHR$ (4): REM  DOS KOMENTOJA VARTEN
20 T$="0123456789ABCDEF": GOSUB 6095: GOTO 130
30 REM  ** DES <--> HES KÄÄNNÖKSET **
40 DE = 0:PO = 1: FOR M = LEN (HE$) TO 1 STEP
- 1: IF M = 0 THEN RETURN
50 FOR G = 0 TO 15: IF MID$ (HE$,M,1) =
MID$ (T$,G + 1,1) THEN 70
60 NEXT G
70 DE = DE + PO * G:PO = PO * 16: NEXT M:
RETURN
80 P = 2:P% = 16: GOTO 100
90 P = 4:P% = 4096
100 HE$ = "": FOR M = 1 TO P:S% = DE / P%
110 HE$ = HE$ + MID$ (T$,S% + 1,1):DE = DE -
S% * P%:P% = P% / 16: NEXT M: RETURN
120 REM  ** TAULUKOIDEN ALUSTUS **
130 DIM OP$(256),O%(256)
140 FOR I = 1 TO 256:OP$(I) = "???":O%(I) =
4: NEXT I: FOR I = 1 TO 38

```

```

150 READ AS$,P:OP$(P + 1) = AS$:O%(P + 1) =
11 - 7 * (I > 8) - 2 * (I > 33) + 10 * (I > 35)
+ (I > 36):: NEXT
160 FOR I = 1 TO 7: READ AS$,P: FOR J = 1 TO
10
170 P1 = P + 4 * (J = 2) - 4 * (J = 3) - 8 *
(J = 6) + 8 * (J = 7) + 12 * (J = 8) + 20 * (J
= 9) + 16 * (J = 10) + 1
175 OP$(P1)=AS$: O%(P1) = J: IF J=3 THEN J = 5
180 NEXT J: NEXT I: FOR I = 1 TO 14: READ
AS$: FOR J = 1 TO 10: READ P: IF P = 0 THEN 190
185 OP$(P + 1) = AS$:O%(P + 1) = J
190 NEXT J: NEXT I
260 HOME
280 INPUT " ALKUOSOITE >";HE$: GOSUB 860:AO =
DE:PC = AO
320 HOME
330 PRINT : PRINT " ASSEMBLY": PRINT
345 IN$ = "":B = 0:DE = PC:HE = 0:Z = 0: GOSUB
90: PRINT "$";HE$: TAB( 20)
360 GET K$
362 IF ASC (K$) < 32 AND NOT (K$ = H$ OR K$
= CHR$( 13)) THEN 360
365 IF K$ = CHR$( 13) THEN 380
367 IF K$ = H$ AND LEN (IN$) > 1 THEN IN$ =
LEFT$( IN$, LEN (IN$) - 1): PRINT H$:: GOTO 360
368 IF K$ = H$ AND LEN (IN$) = 1 THEN IN$ =
"": PRINT H$:: GOTO 360
369 IF K$ = H$ THEN 360
370 PRINT K$::IN$ = IN$ + K$: GOTO 360
380 L = LEN (IN$): IF L < 2 THEN PRINT :
GOTO 345
382 PI$ = LEFT$ (IN$,2): IF PI$ = ".A" THEN
4700
384 IF PI$ = ".D" THEN 4000
386 IF PI$ = ".M" THEN GOSUB 4400: GOTO 330
388 IF PI$ = ".S" THEN GOSUB 4800: GOTO 345
390 IF PI$ = ".L" THEN GOSUB 4900: GOTO 345
392 IF PI$ = ".G" THEN 4600
394 REM ** Tähän voi lisätä uusia komentoja
400 REM
410 REM ** Osoitusmuodon tarkistus **
420 IF L < 3 THEN 3010
430 AS$ = LEFT$ (IN$,3)
440 IF L = 3 THEN OM = 4: GOTO 600
450 I = 4: GOSUB 475: IF L = 4 OR I$ < > " "
THEN 3010
470 I = 5: GOSUB 475: IF I$ < > "#" THEN 480
471 I = I + 1:OM = 1: IF L < I THEN 3000
472 GOSUB 475: IF I$ = "$" THEN HE = 1:I = I
+ 1
473 GOSUB 1000: IF F = 0 OR F > 2 AND HE = 1
OR F > 3 OR L > = I THEN 3000
474 GOTO 600
475 I$ = MID$ (IN$,I,1): RETURN

```

```

480 IF I$ = "$" THEN HE = 1: I = I + 1
485 IF I$ = "(" THEN I = I + 1: GOTO 550
490 GOSUB 1000: IF F = 0 THEN 3000
495 IF AR > 255 THEN Z = 1
500 IF L < I THEN OM = 3 - Z: GOTO 600
510 IF L - I < 1 THEN 3000
520 IF MID$ (IN$, I, 2) = ",X" THEN OM = 8 +
Z: GOTO 600
530 IF MID$ (IN$, I, 2) = ",Y" THEN OM = 13 -
3 * Z: GOTO 600
540 GOTO 3000
550 IF MID$ (IN$, I, 1) = "$" THEN HE = 1: I = I
+ 1
560 GOSUB 1000: IF F = 0 OR F > 5 THEN 3000
570 IF L = I + 2 THEN 590
580 IF MID$ (IN$, I, 1) = ")" THEN OM = 12:
GOTO 600
590 D$ = MID$ (IN$, I, 3): OM = 6 * (D$ = ",X")
+ 7 * (D$ = ",Y"): IF OM = 0 THEN 3000
600 IF (OM = 2 OR OM = 3) AND LEFT$ (AS$, 1)
= "B" AND AS$ < > "BIT" THEN OM = 11
610 FOR I = 1 TO 256: IF AS$ = OP$(I) THEN
IF O%(I) = OM THEN 630
620 NEXT I: GOTO 3000
630 OC = I - 1
790 REM ** TULOSTUS **
800 X = 1: FOR I = 1 TO L + 4: PRINT H$;: NEXT
I: IF OM = 11 THEN 900
810 OM OM GOTO 840, 820, 840, 850, 850, 840,
840, 840, 820, 820, 840, 820, 840
820 X = X + 1: Y% = AR / 256: AR = AR - Y% *
256: MA = PC + 2: DT = Y%: GOSUB 9000: DE = Y%:
GOSUB 80: PRINT HE$; H$; H$;
840 X = X + 1: MA = PC + 1: DT = AR: GOSUB
9000: DE = AR: GOSUB 80: PRINT H$; H$; HE$;
850 MA = PC: DT = OC: GOSUB 9000: DE = OC: GOSUB
80: PRINT H$; H$; H$; H$; H$; HE$: PC = PC + X: GOTO
345
860 IF LEFT$ (HE$, 1) = "$" THEN HE$ =
RIGHT$ (HE$, LEN (HE$) - 1): GOSUB 40: RETURN
870 DE = VAL (HE$): RETURN
900 T = AR - PC - 2: IF T < - 128 OR T > 127
THEN PRINT : PRINT " LIIAN PITKA HYPPIY!": GOTO
345
910 IF T < 0 THEN T = T + 256
920 AR = T: GOTO 810
1000 Y = L - I: X = 4 - HE: FOR F = 0 TO + X *
(X < = Y) + Y * (Y < X): IF L < I THEN 1035
1010 E = ASC (MID$ (IN$, I + F, 1))
1020 IF E > 47 AND E < 58 OR E > 64 AND E <
71 AND HE THEN NEXT
1030 IF HE = 1 AND F < > 2 AND F < > 4 THEN
F = 0
1035 IF F = 0 THEN RETURN

```

```

1040 HE$ = MID$(IN$, I - HE, F + HE): I = I +
F: GOSUB 860: AR = DE: RETURN
3000 PRINT TAB( 30); "VIRHE !": GOTO 345
3010 PRINT TAB( 30); "VIRHE MNE": GOTO 345
3020 PRINT TAB( 30); "VIRHE OSM": GOTO 345
3999 REM
4000 PRINT : PRINT : PRINT "DISASSEMBLY "
4005 GOSUB 4960: IF I = L THEN DC = AD: GOTO
4007
4006 HE$ = N$: GOSUB 860: DC = DE
4007 INPUT " RUUDULLE (R) VAI KIRJOITTIMELLE
(K) >"; F$: PRINT
4009 IF F$ = "R" THEN FA = 0: GOTO 4015
4010 IF F$ < > "K" THEN 4007
4012 PRINT D$; "PR#1": FA = 1
4015 DE = DC: GOSUB 90: PRINT "$"; HE$; " ";
4020 MA = DC: GOSUB 9500: OC = DT: AS$ = OP$(OC
+ 1): OM = O%(OC + 1)
4025 DE = OC: GOSUB 80: OC$ = HE$: IF OM = 11
THEN 4340
4150 IF OM = 4 THEN X = 1: GOTO 4185
4160 MA = DC + 1: GOSUB 9500: DE = DT: GOSUB
80: X$ = HE$: MA = DC + 2: GOSUB 9500: DE = DT:
GOSUB 80: Y$ = HE$
4170 OC$ = OC$ + " " + X$: X = 2
4180 ON OM GOSUB 4210, 4220, 4230, 4230,
4230, 4260, 4270, 4280, 4290, 4300, 4310, 4320,
4330
4185 TB$ = "": FOR Z = 1 TO 20 - LEN
(OC$): TB$ = TB$ + " ": NEXT
4190 PRINT OC$: TB$: AS$: DC = DC + X
4193 GOSUB 7500: IF FLAG = 0 THEN 4015
4194 IF A$ = "X" AND FA = 1 THEN PRINT
D$; "PR# 0": GOTO 330
4195 IF A$ = "X" THEN 330
4198 GET A$
4200 GOTO 4015
4210 AS$ = AS$ + " #" + X$: RETURN
4220 OC$ = OC$ + Y$ + " ": AS$ = AS$ + " $" +
Y$ + X$: X = 3: RETURN
4230 AS$ = AS$ + " $" + X$: RETURN
4260 AS$ = AS$ + " ($" + X$ + ", X)": RETURN
4270 AS$ = AS$ + " ($" + X$ + ")", Y": RETURN
4280 AS$ = AS$ + " $" + X$ + ", X": RETURN
4290 GOSUB 4220: AS$ = AS$ + ", X": RETURN
4300 GOSUB 4220: AS$ = AS$ + ", Y": RETURN
4320 OC$ = OC$ + Y$ + " ": AS$ = AS$ + " ($" +
Y$ + X$ + ")", X = 3: RETURN
4330 AS$ = AS$ + " $" + X$: RETURN
4340 MA = DC + 1: GOSUB 9500: OF = DT: DE = OF:
GOSUB 80: OC$ = OC$ + " " + HE$
4350 DE = DC + OF + 2: IF OF > 127 THEN DE =
DE - 256
4355 GOSUB 90: AS$ = AS$ + " $" + HE$
4360 X = 2: GOTO 4185

```

```

4399 REM
4400 PRINT : PRINT "MONITORI"
4410 GOSUB 4960: IF I = L THEN 3000
4415 HE$ = N$: GOSUB 860: DC = DE
4420 PRINT : DE = DC: GOSUB 90: PRINT "$"; HE$;
4425 MA = DC: GOSUB 9500: DE = DT: GOSUB 80:
PRINT " $"; HE$; "/";
4430 IN$ = ""
4440 GET A$
4445 IF A$ = CHR$ (27) THEN PRINT : RETURN
: REM ESC: POIS MONITORISTA
4450 IF A$ = CHR$ (13) THEN 4510: REM CR :
SEURAAVA OSOITE
4460 IF A$ = CHR$ (9) THEN DC = DC - 1: GOTO
4420: REM TAB: TAVU TAAKSEPAIN
4470 IF A$ = H$ AND LEN (IN$) > 1 THEN IN$ =
LEFT$ (IN$, LEN (IN$) - 1): PRINT H$;; GOTO
4440
4475 IF A$ = H$ AND LEN (IN$) = 1 THEN IN$ =
"": PRINT H$;
4476 IF A$ = H$ THEN GOTO 4440
4480 A = ASC (A$): IF A < 48 AND A < > 36 OR
A > 70 OR A > 57 AND A < 65 THEN 4440
4490 IF LEN (IN$) = 3 THEN 4440
4500 IN$ = IN$ + A$: PRINT A$;; GOTO 4440
4510 IF IN$ = "" THEN 4540
4520 IF LEN (IN$) = 1 THEN 4440
4530 HE$ = IN$: GOSUB 860: IF DE < 256 THEN MA
= DC: DT = DE: GOSUB 9000
4540 DC = DC + 1: GOTO 4420
4598 REM
4599 REM ** .GO **
4600 PRINT : GOSUB 4960: IF I = L THEN MA =
AD: GOSUB 9900: GOTO 330
4605 I = I + 1: GOSUB 475: IF I$ = "$" THEN I
= I + 1: HE = 1
4610 GOSUB 1000: IF F = 0 THEN 3000
4620 MA = AR: GOSUB 9900: GOTO 330
4698 REM
4699 REM ** .ASM **
4700 GOSUB 4960: IF I = L THEN 3000
4720 I = I + 1: GOSUB 475: IF I$ = "$" THEN I
= I + 1: HE = 1
4725 GOSUB 1000: IF F = 0 THEN 3000
4730 PC = AR: PRINT : PRINT : GOTO 345
4798 REM
4799 REM ** SAVE **
4800 GOSUB 4960: IF I = L THEN PRINT TAB(
30): "TIEDOSTONNIMI PUUTTUU!!": RETURN
4802 SA = AD: SL = PC
4805 PRINT : INPUT "ALKUOSOITE: "; HE$: IF HE$
< > "" THEN GOSUB 860: SA = DE
4810 INPUT "LOPPUOSOITE: "; HE$: IF HE$ < > ""
THEN GOSUB 860: SL = DE + 1
4820 PRINT "SAVING..."

```

```

4830 PRINT D$;"BSAVE ";N$;"A";SA;"L";SL -
SA
4840 RETURN
4898 REM
4900 REM ** LOAD **
4910 GOSUB 4960: IF I = L THEN PRINT TAB(
30);"TIEDOSTONNIMI PUUTTUU!!": RETURN
4920 PRINT
4930 PRINT "LOADING..."
4940 PRINT D$;"BLOAD";N$
4950 PRINT : RETURN
4960 FOR I=1 TO L - 1: GOSUB 475: IF I$ = " "
THEN 4980
4970 NEXT I
4980 IF L = I THEN N$ = "": RETURN
4990 N$ = RIGHT$(IN$,L - 1): RETURN
5000 DATA BCC, 144, BCS, 176, BEQ, 240, BMI,
48, BNE, 208, BPL, 16, BVC, 80, BVS, 112, BRK, 0
5010 DATA CLC, 24, CLD, 216, CLI, 88, CLV,
184, DEX, 202, DEY, 136, INX, 232, INY, 200,
NOP, 234
5020 DATA PHA, 72, PHP, 8, PLA, 104, PLP, 40,
RTI, 64, RTS, 96, SEC, 56, SED, 248, SEI, 120
5030 DATA TAX, 170, TAY, 168, TSX, 186, TXA,
138, TXS, 154, TYA, 152, JSR, 32, JMP, 76, JMP,
108
5040 DATA LDX, 182, STX, 150
5050 DATA ADC, 105, AND, 41, CMP, 201, EOR,
73, LDA, 169, ORA, 9, SBC, 233
5060 DATA STA, 0, 141, 133, 0, 0, 129, 145, 149, 157,
153
5070 DATA ASL, 0, 14, 6, 10, 0, 0, 0, 22, 30, 0
5080 DATA BIT, 0, 44, 36, 0, 0, 0, 0, 0, 0, 0
5090 DATA CPX, 224, 236, 228, 0, 0, 0, 0, 0, 0, 0
5100 DATA CPY, 192, 204, 196, 0, 0, 0, 0, 0, 0, 0
5110 DATA DEC, 0, 206, 198, 0, 0, 0, 0, 214, 222, 0
5120 DATA INC, 0, 238, 230, 0, 0, 0, 0, 246, 254, 0
5130 DATA LDX, 162, 174, 166, 0, 0, 0, 0, 0, 0, 190
5140 DATA LDY, 160, 172, 164, 0, 0, 0, 0, 180, 188, 0
5150 DATA LSR, 0, 78, 70, 74, 0, 0, 0, 86, 94, 0
5160 DATA ROL, 0, 46, 38, 42, 0, 0, 0, 54, 62, 0
5170 DATA ROR, 0, 110, 102, 106, 0, 0, 0, 118, 126, 0
5180 DATA STX, 0, 142, 134, 0, 0, 0, 0, 0, 0, 0
5190 DATA STY, 0, 140, 132, 0, 0, 0, 0, 148, 0, 0
6095 REM ** ALOITUSRUUTU **
6100 HOME : PRINT TAB( 8);"** MikroASSEMBLER
**"
6110 PRINT : PRINT
6120 PRINT TAB( 7);"(C) EEP 1985 MB": PRINT
6130 PRINT TAB( 7);"Apple II sovitus"
6140 PRINT TAB( 7);"(C) Aki Korhonen 1985"
6150 PRINT : PRINT
6160 PRINT TAB( 10);" Odota hetkinen ... "
6200 RETURN

```

```
7500 REM ** Onko näppäintä painettu, jos on,  
tallenna tulos A$:iin **  
7510 FLAG = 1  
7520 IF PEEK ( - 16384) > 128 THEN A$ =  
CHR$ ( PEEK ( - 16384) - 128): POKE - 16368,0:  
RETURN  
7530 FLAG = 0: RETURN  
9000 REM ** POKE MA , DT **  
9010 IF MA > 32767 THEN POKE MA - 65536,DT  
9020 IF MA < = 32767 THEN POKE MA,DT  
9030 RETURN  
9500 REM ** PEEK ( MA ) = DT **  
9510 IF MA > 32767 THEN DT = PEEK (MA -  
65536)  
9520 IF MA < = 32767 THEN DT = PEEK (MA)  
9530 RETURN  
9900 REM ** CALL MA **  
9910 IF MA > 32767 THEN CALL MA - 65536  
9920 IF MA < = 32767 THEN CALL MA  
9930 RETURN
```